"Narcos" by "Eleven Dutzend"

Part 1 - Systemblurb

Section 1 - Benutze Extensions

Include German by Narcos version of Team GerX.
After printing the banner text, say "[line break]Credits. Url: credits.html"

Section 2 - Release

Release along with the "Custom-Quixe" interpreter, the source text, a public solution.

Part 2 - Miniextensions

Section 1 - Commenting

After reading a command (this is the ignore comments rule):
	if the player's command matches the regular expression "^#", reject the player's command;

Section 2 - Minitut

After printing the banner text, say "[paragraph break]Newbie-tip: Denk dran mit 'i' in deine Taschen zu gucken und das es 'gib jemand was' gibt. Ausserdem 'b/betrachte etwas' und 'b raum'. Achja, und 'undo'.".

Book 1 - Mauerlaufen

Part 1 - Browsies

The Buch der Browsies is carried by Yourself. The description is "Du blätterst das [printed name] durch. Irgendwie interessant ist ein kleines Bild der Burg. Image: klein_burgbild.jpg".

The Zukuenftiger Brief is carried by Yourself. The printed name is "Zukünftiger Brief".
The description is "Diesen Brief wirst du irgendwann schreiben. IFrame: 32em, erster-brief.html".

The Deutschbuch is carried by Yourself. The description is "Ein Deutschbuch, das du mit dir herumträgst, um gelegentlich Grammatik nachzugucken. IFrame: 25em, deutschbuch.html". Understand "Buch" and "Deutsch" as Deutschbuch.

Part 2 - Rajcac, Turm 2

The Suedostmauer vor Nordostturm is a room. "Du stehst auf der Burgmauer vor einem hohen Turm. Im Südwesten am anderen Ende der Mauer steht noch einer. Im Nordwesten unter der Mauer ist der Burghof, dahinter die innere Burg mit den Ställen davor und dem Brunnen rechts dahinter."
The printed name is "Südostmauer vor Nordturm".

The Orangenbonbons is carried by Yourself. It is plural-named. Understand "Bonbons" as Orangenbonbons.

The Seile is here. It is plural-named. "Einige Seile sind an den Zinnen befestigt und führen hinunter in den Burghof." The description is "Die Seile sind solide an die Zinnen gebunden. Man kann sicher darauf hinunter und hinauf klettern."

Rajcac is here. He is a Person. "Im Osten vor der Turmtür steht  Rajcac, der starken Wächter. Er schaut traurig auf seine Tüte mit Himbeerbonbons und summt ein Liedchen über Orangenbonbons vor sich hin."
Understand "Waechter" as Rajcac.

The Himbeerbonbons is carried by Rajcac. It is plural-named. Understand "Bonbons" as Himbeerbonbons.

Instead of giving the Orangenbonbons to Rajcac:
	say "Rajcac juhuut ganz doll.";
	now Rajcac has the Orangenbonbons;
	if Rajcac has the Himbeerbonbons:
		now Yourself has the Himbeerbonbons;
		say "[line break]Rajcac gibt dir seine Himbeerbonbons."

Part 3 - Biqnap, Turm 1

The Suedostmauer vor Suedwestturm is southwest of Suedostmauer vor Nordostturm. "Du stehst auf der Burgmauer vor einem hohen Turm. Am anderen Ende der Mauer steht noch einer (NO). Unter der Mauer ist der Burghof, dahinter die innere Burg mit den Ställen davor (NW)."
The printed name is "Südostmauer vor Südwestturm".

Bignap is here. He is here. He is a Person. "Im Osten vor der Turmtür steht  Bignap, der starken Wächter. Er schaut traurig auf seine Tüte mit Sahnebonbons und summt ein Liedchen über Himbeerbonbons vor sich hin."
Understand "Waechter" as Bignap.

The Sahnebonbons is carried by Bignap. It is plural-named. Understand "Bonbons" as Sahnebonbons.

Instead of giving the Himbeerbonbons to Bignap:
	say "Bignap juhuut ganz doll. ";
	now Bignap has the Himbeerbonbons;
	if Bignap has the Sahnebonbons:
		now Yourself has the Sahnebonbons;
		say "[line break]Bignap gibt dir seine Sahnebonbons.";
	say "[line break][Gespraech mit Bignap]"

Instead of telling Bignap about something:
	say "[Gespraech mit Bignap]";

To say Gespraech mit Bignap:
	say "Ihr kommt so ins Gespräch über Bonbons, und Bignap verrät dir, das er ganz schrecklich gerne Bonbons ißt, nicht nur beim Karnevals-Mauerlaufen. Aber so die ganz ganz richtigen Bonbons, so die von denen er manchmal träumt, die hat er noch nirgends gefunden. Dabei fallen dir Mamas gute Bummbons ein, und du überlegst, ob du die wohl mittlerweile selbst machen könntest.
	[paragraph break]Tip: Erinnerungen landen auch im Inventory.";
	now yourself has the Erinnerung an Bummbons;

Book 2 - Bummbons 

Erinnerung an Bummbons is a thing. The description is "Bummbons sind ganz toll, auch wenn sie jedesmal anders schmecken. Du erinnerst dich, wie Mama die manchmal gemacht hat.
[paragraph break] Man braucht dafür:
[line break]- Sahne
[line break]- Wasser
[line break]- Wein (ein bischen)
[line break]- Stroh
".

Part 1 - Burghof

The Burghof is below the Suedostmauer. "Du stehst im Burghof. An der Südostmauer hängen einige Seile, an denen man hochklettern könnte. Im Süden ist das Burgtor (TODO), im Norden sind die Ställe und im Nordwesten die Treppen zur Burg (TODO). Im Westen geht die Mauer weiter (TODO). Die Mauern sind von hohen Türmen unterbrochen."

understand "kletter [something]" as going.

[kletter -patch]
instead of exiting when in the Suedostmauer:
	try going down.

Part 2 - Stall

The Stall is north of the Burghof. "Du bist im Stall. Draussen im Süden siehst du den Burghof, und drinnen siehst du einige Pferde."

The Stroh is in the Stall. The special indefinite article is NO ARTICLE.

The Krippe is in the Stall. It is female. It is fixed in place.
The Heu is in the Krippe. The special indefinite article is NO ARTICLE. 

The Pferd Halvar is in the Stall. It is an animal. "Das Pferd Halvar knabbert Heu aus der Krippe und beachtet dich nicht."

The Pferd Kusine is in the Stall. It is an animal. "Das Pferd Kusine knabbert Heu aus der Krippe und beachtet Halvar."

Instead of taking the Heu, say "Kusine guckt dich böse an und nimmt dir das Heu wieder weg."

The etwas Stroh is a thing. The special indefinite article is NO ARTICLE.
Instead of taking the Stroh:
	now yourself carries the etwas Stroh;
	say "Du nimmst etwas von dem Stroh."
	
Part - Kreuz

The Kreuz is in the Stall. It is a supporter. It is not fixed in place.
The Nagel is in the Stall. It is a supporter. It is male. It is fixed in place.

Hanging it on is an action applying to two things.
Understand "haenge [something] an [something]" as hanging it on.
Check hanging it on:
	say "Das geht vielleicht in der Realität." instead.

Instead of hanging the Kreuz on the Nagel when the Kreuz is carried:
	say "Klappt.";
	now the Kreuz is on the Nagel.
	
The zerdoetschter Derwisch is a Person. The printed name is "zerdötschter Derwisch".

Before examining the Kreuz:
	Now the zerdoetschter Derwisch is on the Kreuz.
	
Instead of telling the zerdoetschter Derwisch about something:
	say "Der [noun] freut sich über deine Aufmerksamkeit. Er holt einen Wachturm aus - naja, irgendwoher, presst das Auge an eine Schiessscharte und beginnt mit dogmatisch schauriger Stimme daraus vorzulesen."
	

Part 3 - Ende



	

